Inapplicability of the χ tests

Although this appears obvious, let us construct an explicit example that shows that the χ tests are not appropriate to test the posterior distributions. The example below is completely analytic, and I've chosen parameters to mimic the issue we've seen.

In[1]:=
```
(* Set up a probability distribution *)

$Assumptions = {σ > 0};
```

$$\text{prob}[x\_, \alpha\_, \sigma\_] := \frac{\alpha}{\sqrt{2\pi}} \text{Exp}\left[\frac{-x^2}{2}\right] + \frac{(1-\alpha)}{2\sigma} \text{Exp}[-\text{Abs}[x]/\sigma];$$

In[3]:=
```
(* Analytic, but get Mathematica to check answer *)
```

In[4]:= $\text{Integrate}\left[x^2\,\text{prob}[x, \alpha, \sigma], \{x, -\text{Infinity}, \text{Infinity}\}\right]$

Out[4]= $\alpha + 2\sigma^2 - 2\alpha\sigma^2$

In[5]:= $\text{sig}[\alpha\_, \sigma\_] := \text{Sqrt}\left[\alpha + 2\sigma^2 - 2\alpha\sigma^2\right];$

In[6]:=
```
(* Here are some trivial choices --- this closely mimics the issue we are seeing *)
```

In[7]:=
```
sig[0.8, 1.7]
```

Out[7]=
```
1.39857
```

In[8]:=
```
(* Calculate the χ distribution *)
```

In[9]:=
```
pchi[xt_, α_, σ_] := prob[xt * sig[α, σ], α, σ] sig[α, σ];
```

In[10]:=
```
Plot[{pchi[x, 0.8, 1.7], prob[x, 1, 1]}, {x, -4, 4}, Filling → Axis]
```

Out[10]=



Sensitivity of the KS test

Using the above distribution, we test to see if the KS test could have correctly ruled out the unit normal hypothesis for different numbers of samples.

In[11]:=
```
𝒟 = ProbabilityDistribution[pchi[x, 0.8, 1.7], {x, -∞, ∞}];
```

In[12]:=
```
(* A random sampling from the distribution *)
```

In[13]:=
```
Histogram[RandomVariate[𝒟, 100], {-3, 3, 0.2}, "PDF"]
```

Out[13]=



In[14]:=
```
(* Generate three sets of data, with 100, 160, 600 elements *)
```

In[15]:=
```
data100 = RandomVariate[𝒟, {10^3, 100}];
data160 = RandomVariate[𝒟, {10^3, 160}];
data600 = RandomVariate[𝒟, {10^3, 600}];
```

In[18]:=
```
(* Generate p-value distributions for each of these cases,
comparing to the unit normal *)
```

In[19]:=
```
pvals100 = KolmogorovSmirnovTest[#, NormalDistribution[], "PValue"] & /@ data100;
pvals160 = KolmogorovSmirnovTest[#, NormalDistribution[], "PValue"] & /@ data160;
pvals600 = KolmogorovSmirnovTest[#, NormalDistribution[], "PValue"] & /@ data600;
```

In[22]:=
```
(* Now plot the cumulative distribution in each of these cases *)
```

In[23]:=
```
Histogram[{pvals600, pvals160, pvals100},
 {0, 1, 0.01}, "CDF", ChartStyle → {Red, Green, Blue}]
```

Out[23]=



The Anderson-Darling test is a similar comparison to the cumulative distribution function, except it weights points in the tails more...

In[24]:=
```
(* Overwrite pvals with the tests from the Anderson-Darling tests *)
```

In[25]:=
```
p1vals100 = AndersonDarlingTest[#, NormalDistribution[], "PValue"] & /@ data100;
p1vals160 = AndersonDarlingTest[#, NormalDistribution[], "PValue"] & /@ data160;
p1vals600 = AndersonDarlingTest[#, NormalDistribution[], "PValue"] & /@ data600;
```

In[28]:=
```
Histogram[{p1vals600, p1vals160, p1vals100},
 {0, 1, 0.01}, "CDF", ChartStyle → {Red, Green, Blue}]
```

Out[28]=



In[29]:=
```
(* Compare the 100 object cases *)
```

In[30]:=
```
Histogram[{p1vals100, pvals100}, {0, 1, 0.01}, "CDF", ChartStyle → {Red, Green}]
```

Out[30]=